



DYNAMIC COLOR NEWS is published monthly by DYNAMIC ELECTRONICS, INC., P.O. Box 896, Hartselle, AL 35640, phone (205) 773-2758. Bill Chapple, President; Alene Chapple, Sec. & Treas.; John Pearson, Ph. D. Consultant; Bob Morgan, Ph. D., Consultant.

Entire Contents (c) by DYNAMIC ELECTRONICS INC., 1984. DYNAMIC COLOR NEWS is intended for the private use of our subscribers and purchasers. All rights reserved. Contents of this newsletter may not be copied in whole or in part without written permission from DYNAMIC ELECTRONICS INC. Subscriptions are \$15/yr for U.S.A. & Canada, \$30 other foreign.

The purpose of this newsletter is to provide instruction on Basic & Machine Language programming, Computer theory, operating techniques, computer expansion, plus provide answers to questions from our subscribers.

The submission of questions, operating hints, and solutions to problems to be published in this newsletter are encouraged. All submissions become the property of Dynamic Electronics if the material is used. We reserve the right to edit all material used and not to use material which we determine is unsuited for publication.

All paid subscribers are entitled to discounts of 10% on hardware, and 20% on software manufactured or produced by Dynamic Electronics Inc. plus "specials" mentioned in the newsletter. To receive these discounts use your DCN number which is at the right of your name on the address label. DCN subscribers may obtain a personalized reply to specific computer problems or advice on purchasing equipment. The charge for this service is \$10.

```
*****
*
*   DYNAMIC   COLOR   NEWS   *
*
*       October 1984
*
*   Editor and Publisher
*       Bill Chapple
*
*       Secretary
*       Helen Earls
*
*****
```

#### CONTENTS

Editor's Comments . . . . .	3
Basic Programming . . . . .	3
(Data Handling Tech)	
Machine Language Prog. . . . .	6
(Memory searching)	
96KX Notes . . . . .	8
Operating Hint . . . . .	9

## EDITOR'S COMMENTS

There are a lot of applications where computers control devices. Recently I attended an industrial show in Huntsville. There were all sorts of mechanical devices being controlled by electronic circuits. I am not a machinist but the computer controlled milling machine was quite impressive. Anything that would be needed in a manufacturing plant was on display from fork lifts to office machines. There were several independent telephone companies with displays and the new telephones were really nice. Cleaning supplies, tools, and packaging materials were also displayed.

The office equipment always interests me. A new facimile machine was demonstrated. A facimile machine allows pictures or hand written messages to be sent over the telephone lines. Most machines take from 3 to 5 minutes but these did it in only 1 minute. Two machines were connected together for the demonstration and the copy was very good.

Being an electronics engineer I was naturally more interested in electronic devices. The most impressive device was a device similar to the facimile machine. Let me call it a text reader for discussion. Suppose you have a basic program listed on a piece of paper. You could feed the paper into the text reader and it would read the text, identify the characters, and send the ASCII value of each character to a computer. The computer could handle the information like it was a text file in a word processor. The text could be edited, saved, or printed. The text reader could identify and process characters for 15 different type fonts. How nice it would be to have one of these devices. You could now use a printer for saving your programs. There was only one catch which was the price. The cheapest unit was about \$7,000 and the upper price range was around \$12,000.

This month we are adding a section about using our products. We have received many requests for such information and want to emphasize a particular feature of the product under discussion. It seems that our instructions are never complete enough and this seems to be the case with our competitor's too.

Next month we are starting discussions of how the computer can generate audio or voice. You are familiar with the sounds that can be generated with the sound commands. But how does this work and how can the computer be made to talk? Later we can discuss video. You have probably seen the video music on television but what is video and how does it relate to computers? These are ideas of things we can cover and we would like to have your comments on these subjects.

We have received requests for file handling programs. Included in this issue is a Basic address file program with which you can find information using a telephone number, name, or zip code. Also we are covering finding the memory location for information with machine language subroutines. This will allow us to quickly look through information and be a basis for a high speed sort program in the future.

A word on memory chips. I have noticed that the prices on 256K dynamic RAM chips are falling. Recently I saw some advertised for around \$25 each. This means a set will cost about \$200 which is almost within reach of most people. However to use them with color computers or with any 8bit microprocessor additional circuitry is required.

## BASIC PROGRAMMING DATA HANDLING TECHNIQUES

There are many applications that require data or information to be presented in different formats. Consider an address file that contains names, telephone numbers, Cities, and States. If you are given a telephone

number it would be nice to search through the information in your computer and completely identify the person or firm associated with that telephone number. Let's consider a file with 100 entries. How do we write a basic program that will allow us to obtain information about the entries based on one criteria such as telephone number?

First of all to design a program we need to decide how the information is to be entered into the computer. The information can be contained as part of the program defined, contained within DATA statements, carried within remark statements, or contained within a machine language file. For the last 3 issues of this publication, we showed how to handle information contained within remark statements. For the purpose of this discussion we will use the conventional READ and DATA Statement method of handling the information.

Let's take an example of how we might want information to appear in an address file.

```
John Lewis
3547 Progress St.
New Town, AL 35655
205 533 7652
```

Now let's suppose this information is carried in statement number 160.

```
160 DATA John Lewis,3547 Progress St.,New Town, AL35655, 2055337652
```

Notice we left out spaces between the state and zip code and in the telephone number. It is a disadvantage to leave a space in those locations because they take up extra room in your computer. You can design your program to leave the spaces when you print the information.

In designing a basic program with a lot of information to sort through, it is advantageous to use arrays. To use arrays we need to have a dimension statement near the beginning of the program. So let's use statement 20 for the dimension statement and set up the program to handle 100 address files.

```
20 Z=100: DIM N$(Z), T$(Z), C$(Z), S$(Z), TF$(Z)
25 'N$=NAME: T$=STREET: C$=CITY
30 'S$=STATE & ZIP: TF$=TELEPHONE NUMBER
```

Now let's write a few statements that will read in the data.

```
35 'THIS READS THE DATA
40 FOR J=1 TO 100
45 READ N$(J): IF N$(J)="" THEN 60
50 READ T$(J), C$(J), S$(J), TF$(J)
55 NEXT J
60 NF=J-1: ?"THERE ARE "J" DATA FILES"
```

What do we do next? Go to a menu so we can decide which option we want.

```
65'
70 ?"WHAT DO YOU WANT TO DO?
75 ?"1 DISPLAY ALL INFORMATION
80 ?"2 SEARCH FOR TELEPHONE NUMBER
85 ?"3 SEARCH FOR A NAME
90 ?"4 SEARCH FOR A ZIP CODE
95 INPUT"ENTER THE NUMBER FOR THE OPTION";X
100 ON X GO SUB 200, 400, 600, 800
110 GO TO 75
```

## DISPLAY ALL INFORMATION

Notice that we have broken the problem into several sub sections called subroutines. Now let's write the subroutine starting at statement number 200 to display all information.

```
200 PRINT"THIS DISPLAYS ALL INFORMATION
205 FOR J=1 TO NF
210 GO SUB 250: NEXT J: RETURN
250 PRINTN$(J): PRINTT$(J): PRINTC$(J) + "," + LEFT$(S$(J),2) + " "
    + RIGHT$(S$(J),5)
255 PRINT LEFT$(TF$(J),3)+" "+MID$(TF$(J),4,3)+" "+RIGHT$(TF$(J),4)
257 FOR X=1 TO 500: NEXT X
258 PRINT
260 RETURN
```

Notice we use LEFT\$, MID\$, and RIGHT\$. Let's review these BASIC commands with examples. If X\$="MICROCOMPUTER" then

```
LEFT$(X$,5) ="MICRO"
```

```
MID$(X$, 6, 3) ="COM"
```

```
RIGHT$(X$,5)= "PUTER" Notice that the numbers in the parenthesis
gives the number of characters in the substring for LEFT$ and RIGHT$. For
MID$ the first number, 6 for our example, gives the position counted from
the first character in the string. The second number gives the number of
characters in the string.
```

We are using a subroutine at 250 to print the results of an address file. This will be used if we select other options of the program.

## Search for Telephone Number

If someone calls and leaves you his telephone number then you would like for the computer to display the information for that person. All of the telephone numbers have to be examined until a match is found. Then the information will be displayed.

```
400 ?"THIS SEARCHES FOR A TELEPHONE NUMBER
405 INPUT"THE TELEPHONE NUMBER WITHOUT SPACES";X$
410 FOR J=1 TO NF
415 IF X$=TF$(J) THEN 430 ELSE NEXT J
420?"TELEPHONE NUMBER IS NOT IN THE FILE":RETURN
430 GO SUB 250: RETURN
```

## Search for a Name

What is the telephone number of Jim Jones? This part of the program will search the names and compare your entry with the last name of the name in each file. To make it easier and eliminate the need to remember first names we will just look for the last names. If there are several with the same last name we will list all of them.

```
600 PRINT"THIS SEARCHES FOR A NAME."
605 INPUT"ENTER THE LAST NAME";X$
610 FOR J=1 TO NF
615 L=LEN(N$(J))
620 FOR K=1 TO L
625 A$=MID$(N$(J),K,1)
630 IF A$=" " THEN 640 ELSE NEXT K
640 N$=RIGHT$(N$(J),L-K)
645 IF N$=X$ THEN GO SUB 250 ELSE NEXT J
650 RETURN
```

This program subroutine looks at each character in the name from right to left starting with the one on the right. When a space is detected then the number of characters in the last name in the file is known. We can then pull the last name from the whole name with the RIGHT\$ command. This we compare with the name we entered and if a match occurs then we display the information for that last name.

#### Search for a ZIP Code

This section is similar to the telephone number section except we are looking for a match in zip codes and want to display all entries with the same ZIP codes.

```
800?"THIS DISPLAYS ALL ENTRIES WITH THE SAME ZIP
CODE"
805 INPUT"ENTER THE ZIP CODE"; X$
810 FOR J=1 TO NF
820 ZC$=RIGHT$ (S$(J), 5)
825 IF X$=ZC$ THEN GO SUB 250 ELSE NEXT J
830 RETURN
```

The following are some names we made up to demonstrate the program. They can be entered or you can enter the names of your friends or business associates.

```
1010 DATA Helen Earls,Hwy 20 Tr.Ct.,Decatur, AL35601,2053500183
1020 DATA Virginia Rollins,905 11th ST.Ct.,Decatur,AL35601,2053503001
1030 DATA Dean Smith,543 Court ST.,Dallas,TX44212,9098785234
1040 DATA John Doe,343 Werner Ave.,Smithville, AR34521,7860998455
1050 DATA Faith Jones,54 Longbridge,Kanas,MO44321,7689890076
1060 DATA Hope Norton,2343 Lanier ST.,Newport,IN34565,3099288899
1070 DATA Heddi Waugh,2727 Northfleet RD.,Raytown,MO35600,9742148765
1080 DATA Patti West,Box 545,Zillon,NY21321,3214542111
1090 DATA Will Watson,324 Dogwood Dr.,Wayne,Mich.24310,3139876314
1095 DATA,,,,,
```

Add an empty data statement such as 1095 to indicate an end of data.

## MACHINE LANGUAGE PROGRAMMING

### Memory Searching

There is much interest in handling data in color computers. Suppose you have an inventory program and want to find how many "CHAIRS" you have. If you know the beginning and ending of the memory where the information is located then it is time saving if a machine language program or subroutine exists that will quickly find this information. Elsewhere in this issue we are presenting a program that will handle address files. What we are interested in here is finding information that is located somewhere within our working memory area. For a 32K machine this area might be between 2000 and 30000.

Included is a program that has been checked out. However it was developed on a computer with a cassette and will not work with a disk. Next month we will continue along these lines and present a program that will work with both cassettes and disks.

The program will search and look for a match for a character string up to 8 characters long. It is a basic program with a machine language subroutine that quickly searches memory for a first character match. Basic then takes over and compares the rest of the characters for a match. If no match occurs then the machine language subroutine is recalled and the process continues until a match occurs or the end of memory is reached.

The memory is reserved as follows:

499           Number of characters in the string  
 500-507       Characters in the string  
 508-509       Beginning of the search  
 510-511       End of search  
 512-513       Location of first character match  
 515           End of memory flag  
 4000-4030     Machine language subroutine

#### MACHINE LANGUAGE SUBROUTINE

The machine language subroutine is carried in a DATA Statement at the end of the program. This is the conventional method of handling machine language programs with basic statements. Let's look at the machine language subroutine.

```
4000 LDX E 508  'Put the beginning of memory in 508
4003 CMPXE 510  'Compare X with the end of memory vector in 510
4006 BHS 4025  'Branch if Higher or the Same to 4025
4008 LDA X DIR R+ 'Load A indexed to X and autoincrement X
4010 CMPAE 500  'Compare A with location 500 (First character)
4013 BNE 4008  'GO TO 4008 if not equal
4015 STX E 512  'Save X in location 512
4018 RTS      'Return from the subroutine
4019 LDX E 512  'Load X with the value in 512
4022 BRA 4003  'GO TO 4003
4024 NOP      'Nonoperation or skip this step
4025 LDA I 200  'Load A with the value 200
4027 STA E 515  'Store A in 515 for end of memory flag
4030 RTS      'Return from the subroutine
```

We used our decimal disassembler "DISASM" to disassemble the machine language subroutine. If you have a disassembler then you can use it. Notice that we are linking the information in basic and the machine language subroutine by storing the results in fixed memory locations such as 512 and 515. Location 512 contains the vector for the first character match. We use it to load X with the subroutine. X points to the next byte in memory we want to consider. This byte is compared with the value in location 500. If a match occurs then the X pointer is saved and the RTS command returns to Basic. If no match occurs then the next byte is considered until the end of memory is reached. When starting a "0" is stored in location 515. The machine language subroutine stores the value "200" in location 515 when the end of designated memory occurs.

#### THE BASIC PROGRAM

The program first loads the machine language subroutine into memory locations 4000-4030 using the convention READ and DATA method. It then asks for the string and stores it into memory locations 500 - 507. It computes the length of the string and stores it in location 499.

Then the vectors for the beginning and ending of search are displayed. You can enter a "1" or "2" if you want to change these. If the memory search area is the same as those displayed then enter any other character or press the "ENTER" key to continue.

The program calls the machine language subroutine which searches for a first character match. It returns with the memory location of the first character match stored in location 512. Basic then compares the other characters to see if there is a complete match. If not the machine language subroutine is again called. This process continues until a match occurs or the end of memory is reached.

```

5 'THIS PROGRAM WAS WRITTEN BY BILL CHAPPLE
10 'THIS PROGRAM SEARCHES MEMORY
15 'FOR A CHARACTER MATCH UP TO 8 CHARACTERS
20 'LONG BETWEEN THE BEGINNING AND ENDING LIMITS.
25 '499 NUMBER OF CHARACTERS.
30 '500-507=STRING
35 '508-509=BEGINNING OF SEARCH
40 '510-511=END OF SEARCH
45 '512-513=LOCATION OF PAUSE
50 'ML SUBROUTINE AT 4000
55 'LOAD THE ML PROGRAM
60 FOR J=4000 TO 4030:READX:POKEJ,X:NEXTJ
65 INPUT"STRING";S$:L=LEN(S$)
66 '
67 'STORE THE STRING IN 500-507
68 'STORE THE LENGTH IN 499
70 POKE 499,L:POKE515,0
75 FOR J=1 TO L:A$=MID$(S$,J,1)
80 A=ASC(A$):POKE 499+J,A:NEXT J
85 Y=508:GO SUB 220:BE=Z
90 Y=510:GO SUB 220:EN=Z
95 PRINT"1 BEGINNING OF SEARCH="BE
100 PRINT"2 ENDING OF SEARCH="EN
105 A$=INKEY$:IF A$="" THEN 105
110 IF A$="1" THEN 120 ELSE IF A$="2" THEN 130
115 GO TO 140
120 INPUT"ENTER BEGINNING";X
125 GO SUB 225:POKE 508,MS:POKE509,LS:GO TO 85
130 INPUT"ENDING";X
135 GO SUB 225:POKE 510,MS:POKE 511,LS: GO TO 85
140 EXEC 1900:GOSUB205
145 Y=512:GO SUB 220
150 L=PEEK(499):FOR J=1 TO L
155 M=499+J:N=Z-2+J
160 PRINT"J="J;"M="M;"N="N
165 A=PEEK(M):B=PEEK(N)
170 A$=CHR$(A):B$=CHR$(B)
175 IF A<>B THEN 200
180 NEXT J
185 PRINT"MATCH AT"Z-1
190 INPUT"ENTER A 1 TO CONTINUE";AA
195 IF AA=0 THEN END
200 EXEC 4019: GO SUB 205:GO TO 145
205 IF PEEK(515)=200 THEN PRINT"END OF MEMORY":END
210 RETURN
215 END
220 Z=256*PEEK(Y)+PEEK(Y+1):RETURN
225 MS=INT(X/256):LS=X-256*MS:RETURN
230 DATA 190,1,252,188,1,254,36,17,166,128,177,1,244,38,249,191,2,0,57,190,
2,0,32,235,18,134,200,183,2,3,57

```

## 96KX NOTES

Some of our customers have become a little confused with the 96KX software. Its main purpose is to allow full use of the extra 32K memory page in 64K computers. However there are many useful utilities available which seems to be confusing. These suggestions are for switching information between the two 32K memory pages.

1. EXEC 57370 to make a copy of page 0 to page 1.



2. EXEC 57344 to exchange pages.
3. If the computer hangs up hit the "BREAK" key and type "RUN".
4. EXEC 57701 exchanges Basic program vectors. Sometimes this is undesirable especially when you need the ML vectors. You can access this also by POKING 25,225. Record the value in 25 before you do this so you can return to your program.
5. Steps 1 and 2 are all that is needed if you just want to use all of the memory without the other features.
6. The memory poke and peek utilities in the first menu are very valuable for everyday computer operations.
7. When used with a word processor such as the TW64 you can put the word processor in one bank and use the other for anything else. To force the TW64 software to use only 32K LOAD "U" and add the following line.
 

```
19 GO TO 100
```

 You can return to BASIC from the TW64 menu and then EXEC 57344 to go to the other page.

## OPERATING HINT

If several programs are loaded into your computer you can jump from one to the other by changing the value in location 25. Write down the original value in location 25 so you can return to your original program. You can't edit them unless you change all the vectors in locations 25-28 but they will run. For example when we develop programs we can go to the 96KX software (225) or our "DISASM" assembler (192) in a cartridge by just changing this one value. This leaves the program and its ending vector unchanged while allowing us to do other operations.

### 128K BUTTON

Spectrum Projects has 128K buttons. If you have a 128K upgrade you can contact them if you want a button.

### DYNAMIC ELECTRONICS INC. PRODUCTS

96KX-M Module allows use of all 64K memory	\$49.95
VR-1 Video Reverser	19.95
Uninterrupted Power Source (UPS)	59.95
Second Cartridge Port (new)	44.95

#### Solderless Memory Upgrades

ME-4 Upgrades D & E computers to 64K	89.95
ME-4F Upgrades 285 or F Computers to 64K	79.95
ME-128-64 Upgrades all 64 K to 128K	169.00

SOFTWARE on Tape works on Disk Systems.  
 Extended Basis is not required.

Terminal Program (DYTERM)	14.95
Decimal Assembler (DISASM)	19.95
Multiprogram Manager (MPM) stack 5 PGMS	14.95
Utility Program (UP-1)	14.95

DCN Subscribers take 20% discount for software and 10% for Hardware.

\*\*\*\*\*  
 \* Please sign me up for one year for the DYNAMIC COLOR NEWS SERVICE. I \*  
 \* understand I will receive a monthly news letter, Discounts on DYNAMIC \*  
 \* ELECTRONIC INC. Computer products plus the Individual Reply to my \*  
 \* Computer problems for a special of \$10 each. Also I understand that \*  
 \* there will be no charge for letters printed with answers in the \*  
 \* Newsletter. Cost \$15 USA & Canada, \$30 foreign. \*  
 \* \*  
 \* Name \_\_\_\_\_ Mail payment to \*  
 \* Address \_\_\_\_\_ Dynamic Electronics Inc \*  
 \* City \_\_\_\_\_ P. O. Box 896 \*  
 \* State & Zip \_\_\_\_\_ Hartselle, AL 35640 \*  
 \* Enclosed is a check \_\_\_ \*  
 \* charge to VISA \_\_\_ MC \_\_\_ Number \_\_\_\_\_ Exp. \_\_\_\_\_ \*  
 \* \*  
 \*\*\*\*\*

**DYNAMIC ELECTRONICS INC.**  
 P. O. Box 896 (205) 773-2758  
 Hartselle, AL 35640